APÊNDICE B:  NOTAÇÃO BNF DAS ENTRADAS DO LÉXICO





Abreviaturas utilizadas na sintaxe:





�
1S. = primeira pessoa do singular


2S. = segunda pessoa do singular


3S. = terceira pessoa do singular


1P. = primeira pessoa do plural


2P. = segunda pessoa do plural


3P. = terceira pessoa do plural


3SP. = terceira pessoa do singular e terceira pessoa do plural


A. = aumentativo


ABREV. = abreviatura


ADIT. = aditiva


ADJ. = adjetivo


ADVE. = adversativa


AFIR. = afirmação


ALTER. = alternativa


ART. = artigo


AUX. = auxiliar


BI. = bitransitivo


C. = contração


CAR. = cardinal


CAUS. = causal


CIR-LUG. = circunstância lugar


CIR-TEMP. = circunstância tempo


CIR-MOD. = circunstância modo


COMP. = comparativa


CONC. = concessiva


CONCL. = conclusiva


COND. = condicional


CONFOR. = conformativa


CONJ. = conjunção


CONS. = consecutiva


COORD. = coordenativa


D. = diminutivo


DE. = definido


DEM. = demonstrativo


DUV. = dúvida


ENC. = ênclise


EXPL. = explicativa


F. = feminino


FIN. = final


FRA. = fracionário


FRA-ORD. = fracionário e ordinal


FUT-PRES. = futuro do presente


FUT-PRET. = futuro do pretérito


FUT-SUBJ. = futuro do subjuntivo


GERUN. = gerúndio


I. = artigo indefinido


IMP-AFIRM. = imperativo afirmativo


INDE. = pronome indefinido


INF-PESS. = infinitivo pessoal


INT. = intransitivo


INTE. = intensidade


INTEG. = integrante


INTER. = interrogativo


INT-CAUS. = interrogativo de causa


INT-LUG. = interrogativo de lugar


INT-MOD. = interrogativo de modo


INT-TEMP. = interrogativo de tempo


INV. = invariável


LIG. = ligação


M. = masculino


MESOC. = mesóclise


MUL. = multiplicativo


N. = nulo


NEG. = negação


NOM. = nome próprio


NUM . = numeral


OBL-AT.= pessoal oblíquo átono


OBL-TO.= pessoal oblíquo tônico


ORD. = ordinal


PARTIC. = participio


PL. = plural


POSS. = possessivo


PREF. = prefixo


PREP. = preposição


PRES. = presente


PRES-SUBJ. = presente do subjuntivo


PRET-IMPERF. = pretérito imperfeito


PRET-IMPERF-SUBJ. = pretérito imperfeito do subjuntivo


PRET-M-Q-P. = pretérito mais que perfeito


PRET-PERF. = pretérito perfeito


PRON. = pronome


PRONOM. =pronominal


PROPOR. = proporcional


RET. = pessoal reto


REL. = relativo


REFL. = reflexivo


S. = substantivo


SI. = singular


SIGL. = sigla


SU. = superlativo


SUBORD. = subordinativa


TD. = transtivo direto


TEMP. = temporal


TI. = transtivo indireto


TRAT. = tratamento


V. = verbo


2G. = dois gênero


2N. = dois número

             ::=	Significa “é definido como”.


	|	Significa “ou”.


	{ }	Significa que o conteúdo pode ser repetido nenhuma ou várias vezes.


	< >	Significa que o conteúdo é uma construção sintática BNF.





<verbete> ::= <palavra> =< <classificação> { # <classificação>} >


<abreviatura> ::= ABREV.


<adjetivo> ::= ADJ. <gênero-2> <número-2> <grau-2> <regência> <regra-de-derivação>


<advérbio> ::= ADV. <tipo-do-advérbio> <grau-2>


<artigo> ::= ART. <gênero-1> <número-1> <tipo-do-artigo> <regra-de-derivação>


<canônica> ::= [ <palavra> ]


<caracter> ::= a..z | á | é | í | ó | ú | à | â | ê | ô | ã | õ | ç | ü | - | ‘ | ª | º


<classe> ::= <substantivo> | <artigo> | <preposição> | <numeral> | <pronome> | <verbo> | 	        <adjetivo> | <conjunção> | <advérbio> | <nome-próprio> | <prefixo> | <sigla> | <abreviatura>


<classificação> ::= <classe> <canônica> <flag-composição>


<colocação-pronominal> ::= N. | ENC. | MESOC.


<conjugação> ::= [ <tempo> <pessoa-1> | <forma-nominal>


		 {<tempo> <pessoa-1> | <forma-nominal>} ]


<conjunção> ::= CONJ. <tipo-da-conjunção>


<conjunções-coordenadas> ::= [ <tipos-de-conj-coord> {<tipos-de-conj-coord} ]


<conjunções-subordinadas> ::= [ <tipos-de-conj-subord> {<tipos-de-conj-subord>} ]


<contração> ::= C. <lista-de-palavras>


<dígito> ::= 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9


<flag-composição> ::= <dígito>


<forma-nominal> ::= GERUN. | PARTIC. <gênero-2> <número-2>


<gênero-1> ::= M. | F.


<gênero-2> ::= M. | F. | 2G.


<gênero-3> ::= M. | F. | 2G. | INV.


<grau-1> ::= N. | A. | D.


<grau-2> ::= N. | A. | D. | SU.


<lista-de-palavras> ::= [ <palavra> . {<palavra> . } ]


<nome-próprio> ::= NOM. <gênero-2> <número-2>


<numeral> ::= NUM. <gênero-3> <número-3> <tipo-do-numeral> <regra-de-derivação>


<número-1> ::= SI. | PL.


<número-2> ::= SI. | PL. | 2N.


<número-3> ::= SI. | PL. | 2N. | INV.


<palavra> ::= <caracter> {<caracter>}


<pessoa-1> ::= EU. | TU. | ELE. | NOS. | VOS. | ELES.


<pessoa-2> ::= 1S. | 2S. | 3S. | 1P. | 2P. | 3P. | 3SP. | N.


<predicação> ::= [ <tipo-da-predicação> {<tipo-da-predicação>} ]


<prefixo> ::= PREF.


<preposição> ::= PREP. <contração>





<pronome> ::= PRON. <gênero-3> <número-3> <tipo-do-pronome> <pessoa-2>


		 <regra-de-derivação> <contração>


<regência> ::= <lista-de-palavras>


<regra-de-derivação> ::= <dígito> | ?


<sigla> ::= SIGL.


<substantivo> ::= S. <gênero-2> <número-3> <grau-1> <regência> <regra-de-derivação>


<tempo> ::= PRES. | PRET-IMPERF. | PRET-PERF. | PRET-M-Q-P. | FUT-PRES. |


	       FUT-PRET. | PRES-SUBJ. | PRET-IMPERF-SUBJ. | FUT-SUBJ. |


	       IMPER-AFIRM. | INF-PESS.


<tipo-da-predicação> ::= INT. | TD. | TI. | BI. | LIG. | AUX. | PRONOM.


<tipo-do-advérbio> ::= CIR-LUG. | CIR-TEMP. | CIR-MOD. | NEG. | DUV. | INT. |


		         AFIR. | INT-LUG. | INT-TEMP. | INT-MOD. | INT-CAUS.


<tipo-do-artigo> ::= DE. | I.


<tipo-da-conjunção> ::= [ COORD. <conjunções-coordenadas> |


			 SUBORD. <conjunções-subordinadas>


			 { COORD. <conjunções-coordenadas> |


			  SUBORD. <conjunções-subordinadas> } ]


<tipo-do-numeral> ::= ORD. | MUL. | FRA. | FRA-ORD. | CAR.


<tipo-do-pronome> ::= TRAT. | RET. | OBL-AT. | OBL-TO. | POSS. | DEM. | INDE. | 			                        INTE. | REL. | REFL.


<tipos-de-conj-coord> ::= ADIT. | ADVE. | ALTER. | CONCL. | EXPL.


<tipos-de-conj-subord> ::= INTE. | CAUS. | COMP. | CONC. | COND. | CONS. | FIN.


			  TEMP. | PROPOR. | CONFOR.


<verbo> ::= V. <predicação> <conjugação> <colocação-pronominal> <regência>

